



TSwap Protocol Audit Report

Version 1.0

AlexGer

August 28, 2025

TSwap Protocol Audit Report

AlexGer

August 28, 2025

Prepared by: AlexGer Lead Auditors:

- Aleksei Gerasev

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High
 - * [H-1] Incorrect fee calculation in `TSwapPool::getInputAmountBasedOnOutput` causes protocol to take too many tokens from users, resulting in lost fees
 - * [H-2] Lack of slippage protection in `TSwapPool::swapExactOutput` causes user to potentially receive way fewer tokens
 - * [H-3] `TSwapPool::sellTokens` mismatches input and output tokens causing users to receive the incorrect amount of tokens

- * [H-4] In `TSwapPool::_swap` the extra tokens given to users after every `swapCount` breaks the protocol invariant of $x * y = k$
- Medium
 - * [M-1] `TSwapPool::deposit` is missing deadline check causing transactions to complete even after the deadline
 - * [M-2] Rebase, fee-on-transfer, and ERC-777 tokens break protocol invariant
- Low
 - * [L-1] `TSwapPool::LiquidityAdded` event has parameteres out of order
 - * [L-2] Default value returned by `TSwapPool::swapExactInput` results in incorrect return value given
- Info
 - * [I-1] Lack of zero address validation
 - * [I-2] `PoolFactory::PoolFactory__PoolDoesNotExist` is not used and should be removed
 - * [I-3] `PoolFactory::createPool` should use `.symbol()` instead of `.name()`
 - * [I-4] Event `TSwapPool::Swap` is missing `indexed` fields
 - * [I-5] Function `TSwapPool::swapExactInput` should be `external` instead of `public` because it is not used internally
 - * [I-6] Missing natspec for the `TSwapPool::swapExactInput` function
 - * [I-7] Missing natspec for the `deadline` parameter in the `TSwapPool::swapExactOutput` function
 - * [I-8] Using magic numbers instead of constants
 - * [I-9] It's better to follow the CEI (Checks, Effects, and Interactions) principle in the else branch of the `TSwap::deposit` function
- Gas
 - * [G-1] Function `TSwapPool::_isUnknown` can be simplified
 - * [G-2] Local variable `poolTokenReserves` in the function `TSwapPool::deposit` is not used

Protocol Summary

TSWAP is an constant-product AMM that allows users permissionlessly trade WETH and any other ERC20 token set during deployment. Users can trade without restrictions, just paying a tiny fee in each swapping operation. Fees are earned by liquidity providers, who can deposit and withdraw liquidity at any time.

Disclaimer

The AlexGer team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

- Commit Hash: e643a8d4c2c802490976b538dd009b351b1c8dda

Scope

```
1 ./src/  
2 #-- PoolFactory.sol  
3 #-- TSwapPool.sol
```

- Solc Version: 0.8.20
- Chain(s) to deploy contract to: Ethereum
- Tokens:
 - Any ERC20 token

Roles

- Liquidity Providers: Users who have liquidity deposited into the pools. Their shares are represented by the LP ERC20 tokens. They gain a 0.3% fee every time a swap is made.
- Users: Users who want to swap tokens.

Executive Summary

Issues found

Severity	Number of issues found
High	4
Medium	2
Low	2
Info	9
Gas	2
Total	19

Findings

High

[H-1] Incorrect fee calculation in TSwapPool::getInputAmountBasedOnOutput causes protocol to take too many tokens from users, resulting in lost fees

Description: The `getInputAmountBasedOnOutput` function is intended to calculate the amount of tokens a user should deposit given an amount of tokens of output tokens. However, the function currently miscalculates the resulting amount. When calculating the fee, it scales the amount by 10_000 instead of 1_000.

Impact: Protocol takes more fees than expected from users.

Proof of Concept:

Place this code in `TSwapPool.t.sol`

```
1     function testCollectFeesWrong() public {
2         uint256 wethAmount = 100e18;
3         uint256 poolTokenAmount = 100e18;
4
5         vm.startPrank(liquidityProvider);
6         weth.approve(address(pool), wethAmount);
7         poolToken.approve(address(pool), poolTokenAmount);
8         pool.deposit({
9             wethToDeposit: wethAmount,
10            minimumLiquidityTokensToMint: 0,
11            maximumPoolTokensToDeposit: poolTokenAmount,
12            deadline: uint64(block.timestamp)
13        });
14
15        vm.stopPrank();
16
17        address someUser = makeAddr("someUser");
18        uint256 initialUserPoolTokenAmount = 11e18;
19        uint256 outputWethUserAmount = 1 ether;
20        poolToken.mint(someUser, initialUserPoolTokenAmount);
21
22        vm.startPrank(someUser);
23        poolToken.approve(address(pool), type(uint256).max);
24
25        // in the pool now 1:1, so the user should get 1 WETH and pay
26        // ~1 poolToken
27        pool.swapExactOutput({
28            inputToken: poolToken,
29            outputToken: weth,
30            outputAmount: outputWethUserAmount,
31            deadline: uint64(block.timestamp)
32        });
33
34        vm.stopPrank();
35
36        uint256 userPoolTokenBalanceAfter = poolToken.balanceOf(
37            someUser);
38        console.log("userPoolTokenBalanceAfter",
39            userPoolTokenBalanceAfter);
40
41        uint256 userSpentPoolToken = initialUserPoolTokenAmount -
42            userPoolTokenBalanceAfter;
43        console.log("userSpentPoolToken", userSpentPoolToken);
44        // user spent more than 10 poolTokens
45        assertGt(userSpentPoolToken, 10e18);
46    }
```

Recommended Mitigation:

```
1     function getInputAmountBasedOnOutput(
```

```
2      uint256 outputAmount,  
3      uint256 inputReserves,  
4      uint256 outputReserves  
5  )  
6      public  
7      pure  
8      revertIfZero(outputAmount)  
9      revertIfZero(outputReserves)  
10     returns (uint256 inputAmount)  
11  {  
12     return  
13 -      (((inputReserves * outputAmount) * 10000) /  
14 +      ((inputReserves * outputAmount) * 1000) /  
15      ((outputReserves - outputAmount) * 997);  
16  }
```

[H-2] Lack of slippage protection in TSwapPool::swapExactOutput causes user to potentially receive way fewer tokens

Description: The `swapExactOutput` function does not include any sort of slippage protection. This function is similar to what is done in `TSwapPool::swapExactInput`, where the function specifies a `minOutputAmount`, the `swapExactOutput` function should specify a `maxInputAmount`.

Impact: If market conditions change before the transaction processes, the user could get a much worse swap.

Proof of Concept:

1. The price of WETH right now is 1,000 USDC
2. User inputs a `swapExactOutput` looking for 1 WETH
 1. inputToken = USDC
 2. outputToken = WETH
 3. outputAmount = 1
 4. deadline = whatever
3. The function does not offer a maxInput amount
4. As the transaction is pending in the mempool, the market changes! And the price moves HUGE
-> 1 WETH is now 10,000 USDC. 10x more than the user expected
5. The transaction completes, but the user sent the protocol 10,000 USDC instead of the expected 1,000 USDC

Proof of Code

Place this in `TSwapPool.t.sol`

```
1      function testNoSlippageProtection() public {
2          uint256 initialLiquidity = 10e18;
3
4          vm.startPrank(liquidityProvider);
5          weth.approve(address(pool), initialLiquidity);
6          poolToken.approve(address(pool), initialLiquidity);
7          pool.deposit({
8              wethToDeposit: initialLiquidity,
9              minimumLiquidityTokensToMint: 0,
10             maximumPoolTokensToDeposit: initialLiquidity,
11             deadline: uint64(block.timestamp)
12         });
13         vm.stopPrank();
14
15         address swapper = makeAddr("swapper");
16         uint256 swapperAmount = 9e18;
17         weth.mint(swapper, swapperAmount);
18
19         vm.startPrank(swapper);
20         weth.approve(address(pool), type(uint256).max);
21         pool.swapExactInput({
22             inputToken: weth,
23             inputAmount: swapperAmount,
24             outputToken: poolToken,
25             minOutputAmount: 0,
26             deadline: uint64(block.timestamp)
27         });
28         vm.stopPrank();
29
30         // ~4.7 pool token after swap
31         uint256 poolTokenBalanceAfter = poolToken.balanceOf(swapper);
32         console.log("poolTokenBalanceAfter", poolTokenBalanceAfter);
33
34         uint256 userAmount = 100e18;
35         weth.mint(user, userAmount);
36
37         vm.startPrank(user);
38         weth.approve(address(pool), type(uint256).max);
39         // and here we have a huge slippage and the worst course
40         uint256 userPaid = pool.swapExactOutput({
41             inputToken: weth,
42             outputToken: poolToken,
43             outputAmount: 1e18,
44             deadline: uint64(block.timestamp)
45         });
46         vm.stopPrank();
47
48         // user paid much more than the initial liquidity!
49         console.log("userPaid for 1 WETH", userPaid);
50         assertGt(userPaid, initialLiquidity);
```



```
51     }
```

Recommended Mitigation: We should include a `maxInputAmount` so the user only has to spend up to a specific amount, and can predict how much they will spend on the protocol.

```
1  function swapExactOutput(  
2      IERC20 inputToken,  
3  +    uint256 maxInputAmount,  
4      IERC20 outputToken,  
5      uint256 outputAmount,  
6      uint64 deadline  
7  )  
8      public  
9      revertIfZero(outputAmount)  
10     revertIfDeadlinePassed(deadline)  
11     returns (uint256 inputAmount)  
12  {  
13     uint256 inputReserves = inputToken.balanceOf(address(this));  
14     uint256 outputReserves = outputToken.balanceOf(address(this));  
15  
16     inputAmount = getInputAmountBasedOnOutput(  
17         outputAmount,  
18         inputReserves,  
19         outputReserves  
20     );  
21  
22  +     if (inputAmount > maxInputAmount) {  
23  +         revert();  
24  +     }  
25     _swap(inputToken, inputAmount, outputToken, outputAmount);  
26 }
```

[H-3] TSwapPool::sellTokens mismatches input and output tokens causing users to receive the incorrect amount of tokens

Description: The `sellPoolTokens` function is intended to allow users to easily sell pool tokens and receive WETH in exchange. Users indicate how many tokens they're willing to sell in the `poolTokenAmount` parameter. However, the function currently miscalculates the swapped amount.

This is due to the fact that the `swapExactOutput` function is called, whereas the `swapExactInput` function is the one that should be called. Because users specify the exact amount of input tokens, not output.

Impact: Users will swap the wrong amount of tokens, which is a severe disruption of protocol functionality.

Proof of Concept:

Proof of Code

Place this in TSwapPool.t.sol

```
1 function testSellPoolTokensSwapsWrongAmount(  
2     uint256 poolTokenAmount  
3 ) public {  
4     uint256 initialLiquidity = 100e30;  
5     poolTokenAmount = bound(poolTokenAmount, 0, initialLiquidity);  
6     weth.mint(liquidityProvider, initialLiquidity);  
7     poolToken.mint(liquidityProvider, initialLiquidity);  
8  
9     vm.startPrank(liquidityProvider);  
10    weth.approve(address(pool), initialLiquidity);  
11    poolToken.approve(address(pool), initialLiquidity);  
12    pool.deposit({  
13        wethToDeposit: initialLiquidity,  
14        minimumLiquidityTokensToMint: 0,  
15        maximumPoolTokensToDeposit: initialLiquidity,  
16        deadline: uint64(block.timestamp)  
17    });  
18    vm.stopPrank();  
19  
20    address swapper = makeAddr("swapper");  
21    poolToken.mint(swapper, poolTokenAmount);  
22  
23    //try to sell poolToken for WETH with sellPoolTokens  
24    vm.startPrank(swapper);  
25    poolToken.approve(address(pool), type(uint256).max);  
26    // this should revert with any amount because inside  
27    // sellPoolTokens we call swapExactOutput  
28    vm.expectRevert();  
29    pool.sellPoolTokens(poolTokenAmount);  
30    vm.stopPrank();  
31 }
```

Recommended Mitigation: Consider changing the implementation to use `swapExactInput` instead of `swapExactOutput`. Note that this would also require changing the `sellPoolTokens` function to accept a new parameter (ie `minWethToReceive` to be passed to `swapExactOutput`)

```
1 function sellPoolTokens(  
2     uint256 poolTokenAmount,  
3     +     uint256 minWethToReceive  
4 ) external returns (uint256 wethAmount) {  
5     return  
6     -     swapExactOutput(  
7     +     swapExactInput(  
8         i_poolToken,  
9     +     poolTokenAmount
```

```
10         i_wethToken,  
11     +     minWethToReceive,  
12     -     poolTokenAmount,  
13         uint64(block.timestamp)  
14     );  
15 }
```

Additionally, it might be wise to add a deadline to the function, as there is currently no deadline (MEV).

[H-4] In TSwapPool : : _swap the extra tokens given to users after every swapCount breaks the protocol invariant of $x * y = k$

Description: The protocol follows a strict invariant of $x * y = k$. Where:

- x : The balance of the pool token
- y : The balance of WETH
- k : The constant product of the two balaances

This means, that whenever the balances change in the protocol, the ratio between the two amounts should remain constant, hence the k . However, this is broken due to the extra incentive in the `_swap` function. Meaning that over time the protocol funds will be drained.

Impact: A user could maliciously drain the protocol of funds by doing a lot of swaps and collecting the extra incentive given out by the protocol.

Most simply put, the protocol's core invariant is broken.

The following block of code is responsible for the issue.

```
1 swap_count++;  
2 if (swap_count >= SWAP_COUNT_MAX) {  
3     swap_count = 0;  
4     outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000);  
5 }
```

Proof of Concept:

1. A user swaps 10 times, and collects the extra incentive of 1_000_000_000_000_000_000 tokens
2. That user continues to swap untill all the protocol funds are drained

Proof of Code

Place the following into `TSwapPool.t.sol`

```
1     function testInvariantBroken() public {
2         vm.startPrank(liquidityProvider);
3         weth.approve(address(pool), 100e18);
4         poolToken.approve(address(pool), 100e18);
5         pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6         vm.stopPrank();
7
8         uint256 wethAmount = 1e17;
9
10        vm.startPrank(user);
11        poolToken.approve(address(pool), type(uint256).max);
12        poolToken.mint(user, 100e18);
13
14        for (uint256 i = 1; i <= 9; i++) {
15            pool.swapExactOutput(
16                poolToken,
17                weth,
18                wethAmount,
19                uint64(block.timestamp)
20            );
21        }
22
23        int256 startingY = int256(weth.balanceOf(address(pool)));
24        int256 expectedDeltaY = int256(wethAmount);
25
26        pool.swapExactOutput(
27            poolToken,
28            weth,
29            wethAmount,
30            uint64(block.timestamp)
31        );
32
33        vm.stopPrank();
34
35        uint256 endingY = weth.balanceOf(address(pool));
36        int256 actualDeltaY = int256(endingY) - int256(startingY);
37        assertEq(actualDeltaY, expectedDeltaY);
38    }
```

Recommended Mitigation: Remove the extra incentive mechanism. If you want to keep this in, we should account for the change in the $x * y = k$ protocol invariant. Or, we should set aside tokens in the same way we do with fees.

```
1 - swap_count++;
2 - if (swap_count >= SWAP_COUNT_MAX) {
3 -     swap_count = 0;
4 -     outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000)
5 - ;
6 - }
```

Medium

[M-1] TSwapPool::deposit is missing deadline check causing transactions to complete even after the deadline

Description: The `deposit` function accepts a deadline parameter, which according to the documentation is “The deadline for the transaction to be completed by”. However, this parameter is never used. As a consequence, operations that add liquidity to the pool might be executed at unexpected times, in market conditions where the deposit rate is unfavorable.

Impact: Transactions could be sent when market conditions are unfavorable to deposit, even when adding a deadline parameter.

Proof of Concept: The `deadline` parameter is unused.

Recommended Mitigation: Consider making the following change to the function.

```
1     function deposit(  
2         uint256 wethToDeposit,  
3         uint256 minimumLiquidityTokensToMint,  
4         uint256 maximumPoolTokensToDeposit,  
5         uint64 deadline  
6     )  
7     external  
8 +     revertIfDeadlinePassed(deadline)  
9     revertIfZero(wethToDeposit)  
10    returns (uint256 liquidityTokensToMint)
```

[M-2] Rebase, fee-on-transfer, and ERC-777 tokens break protocol invariant

Description: The TSwapPool contract assumes that the `i_wethToken` and `i_poolToken` are standard ERC-20 tokens, where the amount transferred in or out matches the amount specified in `safeTransfer` and `safeTransferFrom` calls. However, it does not account for non-standard ERC-20 tokens, such as:

- Rebase Tokens: Tokens that automatically adjust their total supply or individual balances (e.g., Ampleforth), causing the contract’s recorded balance (`balanceOf(address(this))`) to differ from expected reserves.
- Fee-on-Transfer Tokens: Tokens that deduct a fee during transfers (e.g., PAXG), so the amount received by the contract or user is less than the specified amount.
- ERC-777 Tokens: Tokens with callback hooks (e.g., `tokensReceived`), which could allow reentrancy or unexpected behavior during transfers.

Impact:

- **Invariant Breakage:** The AMM's constant product invariant ($x * y = k$) is violated if actual token amounts differ from expected, leading to incorrect pricing, liquidity calculations, or swap outputs. This could allow arbitrageurs to drain reserves or cause losses for liquidity providers.
- **Financial Loss:** Users may receive fewer tokens than expected (e.g., in swaps or withdrawals) or deposit more than intended, reducing their returns.
- **Reentrancy Risk (ERC-777):** If `i_wethToken` or `i_poolToken` is an ERC-777 token, its callbacks could reenter the contract, potentially manipulating state or causing unexpected behavior (though `SafeERC20` mitigates some risks by using non-reentrant functions).
- **Protocol Disruption:** Repeated transactions with such tokens could compound errors, making the pool unusable or unfairly skewed.

Proof of Concept: Consider a fee-on-transfer token as `i_poolToken` with a 1% transfer fee:

In deposit, a user calls with `wethToDeposit = 1000` and `maximumPoolTokensToDeposit = 1000`. The contract calculates `poolTokensToDeposit = 1000` (via `getPoolTokensToDepositBasedOnWeth`). During `_addLiquidityMintAndTransfer`, `i_poolToken.safeTransferFrom(msg.sender, address(this), 1000)` is called. Due to a 1% fee, only 990 pool tokens are received, but the contract assumes 1000. The invariant $k = \text{wethReserves} * \text{poolTokenReserves}$ is calculated based on 1000 pool tokens, not 990, leading to incorrect liquidity token minting and future pricing errors.

For rebase tokens, if `i_poolToken` rebases down after a deposit, `i_poolToken.balanceOf(address(this))` decreases unexpectedly, breaking the invariant in subsequent swaps. For ERC-777, a malicious `tokensReceived` callback could attempt reentrancy, though `SafeERC20` reduces this risk by avoiding direct call.

Proof of Code

Mock fee-on-transfer token

```
1      contract FeeToken is ERC20 {
2          constructor() ERC20("FeeToken", "FEE") {
3              _mint(msg.sender, 1_000_000);
4          }
5          function transfer(address recipient, uint256 amount) public
6              override returns (bool) {
7              uint256 fee = amount / 100; // 1% fee
7              _burn(msg.sender, fee);
8              _transfer(msg.sender, recipient, amount - fee);
9              return true;
10         }
11         function transferFrom(address sender, address recipient,
12             uint256 amount) public override returns (bool) {
13             uint256 fee = amount / 100;
13             _burn(sender, fee);
14             _transfer(sender, recipient, amount - fee);
```

```
15         return true;
16     }
17 }
```

Place this test and mock fee-on-transfer token in `TSwapPool.t.sol`

```
1     function testFeeOnTransferBreaksInvariant() public {
2         FeeToken feeToken = new FeeToken();
3
4         uint256 wethToDeposit = 1e18;
5         uint256 maxPoolTokens = 1e18;
6
7         pool = new TSwapPool(address(feeToken), address(weth), "TokenA",
8             "LA");
9
10        feeToken.mint(user, maxPoolTokens);
11
12        vm.startPrank(user);
13        feeToken.approve(address(pool), type(uint256).max);
14        weth.approve(address(pool), type(uint256).max);
15        pool.deposit(wethToDeposit, 0, maxPoolTokens, uint64(block.
16            timestamp));
17        vm.stopPrank();
18
19        // Check reserves
20        uint256 wethReserves = 1e18; // Mocked
21        uint256 feeTokenReserves = feeToken.balanceOf(address(pool));
22        assertEq(feeTokenReserves, 0.99e18, "Expected tokens after fee"
23            );
24
25        // Invariant check: k = wethReserves * feeTokenReserves
26        // Expected k (if no fee): 1e18 * 1e18 = 1e36
27        // Actual k: 1e18 * 0.99e18 = 0.99e36
28        uint256 actualK = wethReserves * feeTokenReserves;
29        assertEq(actualK, 0.99e36, "Invariant broken due to fee-on-
30            transfer");
31    }
```

Recommended Mitigation:

1. Measure balance changes: update transfer functions to measure `balanceOf(address(this))` before and after transfers to determine actual amounts received or sent.

```
1     function _addLiquidityMintAndTransfer(
2         uint256 wethToDeposit,
3         uint256 poolTokensToDeposit,
4         uint256 liquidityTokensToMint
5     ) private {
6         _mint(msg.sender, liquidityTokensToMint);
7         emit LiquidityAdded(msg.sender, poolTokensToDeposit,
8             wethToDeposit);
9     }
```

```
8
9 +     uint256 wethBalanceBefore = i_wethToken.balanceOf(address(this)
10 + );
11 +     uint256 poolTokenBalanceBefore = i_poolToken.balanceOf(address(
12 + this));
13
14     i_wethToken.safeTransferFrom(msg.sender, address(this),
15     wethToDeposit);
16     i_poolToken.safeTransferFrom(
17     msg.sender,
18     address(this),
19     poolTokensToDeposit
20 );
21
22 +     uint256 wethReceived = i_wethToken.balanceOf(address(this)) -
23 + wethBalanceBefore;
24 +     uint256 poolTokensReceived = i_poolToken.balanceOf(address(this)
25 + )) - poolBalanceBefore;
26 +     require(wethReceived >= wethToDeposit, "WETH transfer failed or
27 + fee-on-transfer");
28 +     require(poolTokensReceived >= poolTokensToDeposit, "Pool token
29 + transfer failed or fee-on-transfer");
30 }
```

Apply similar logic in `_swap`, `withdraw`, etc.

2. Restrict Token Types: add a check in the constructor or a configuration function to ensure `i_wethToken` and `i_poolToken` are standard ERC-20 tokens (e.g., reject known rebase/ERC-777 tokens via a registry or interface check)
3. Disable ERC-777 Callbacks: explicitly use `SafeERC20`'s non-callback methods or add reentrancy guards (though `SafeERC20` already mitigates much of this).

Low

[L-1] TSwapPool::LiquidityAdded event has parameteres out of order

Description: When the `LiquidityAdded` event is emitted in the `TSwapPool::_addLiquidityMintAndTransfer` function, it logs values in an incorrect order. The `poolTokensToDeposit` value should go in the third parameter position, whereas the `wethToDeposit` value should go second.

Impact: Event emission is incorrect, leading to off-chain functions potentially malfunctioning.

Recommended Mitigation:

```
1 - emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit)
;
;
```



```
2 + emit LiquidityAdded(msg.sender, wethToDeposit, poolTokensToDeposit);
```

[L-2] Default value returned by TSwapPool::swapExactInput results in incorrect return value given

Description: The `swapExactInput` function is expected to return the actual amount of tokens bought by the caller. However, while it declares the named return value `output` it is never assigned a value, nor uses an explicit return statement.

Impact: The return value will always be 0, giving incorrect information to the caller.

Proof of Concept:

Place this code in `TSwapPool.t.sol`

```
1 function testSwapExactInputAlwaysReturnsZero() public {
2     vm.startPrank(LiquidityProvider);
3     weth.approve(address(pool), 100e18);
4     poolToken.approve(address(pool), 100e18);
5     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6     vm.stopPrank();
7
8     vm.startPrank(user);
9     uint256 initialUserWethBalance = weth.balanceOf(user);
10    poolToken.approve(address(pool), 10e18);
11    uint256 userSpentPoolToken = pool.swapExactInput(
12        poolToken,
13        10e18,
14        weth,
15        0,
16        uint64(block.timestamp)
17    );
18    vm.stopPrank();
19
20    uint256 userSpentWeth = weth.balanceOf(user) -
21        initialUserWethBalance;
22    assertNotEq(userSpentWeth, 0);
23    assertEq(userSpentPoolToken, 0);
24 }
```

Recommended Mitigation:

```
1 function swapExactInput(
2     IERC20 inputToken,
3     uint256 inputAmount,
4     IERC20 outputToken,
5     uint256 minOutputAmount,
```

```
6         uint64 deadline
7     )
8     public
9     revertIfZero(inputAmount)
10    revertIfDeadlinePassed(deadline)
11    returns (
12        uint256 output
13    )
14    {
15        uint256 inputReserves = inputToken.balanceOf(address(this));
16        uint256 outputReserves = outputToken.balanceOf(address(this));
17
18 -        uint256 outputAmount = getOutputAmountBasedOnInput(
19 +        output = getOutputAmountBasedOnInput(
20            inputAmount,
21            inputReserves,
22            outputReserves
23        );
24
25 -        if (outputAmount < minOutputAmount) {
26 +        if (output < minOutputAmount) {
27 -            revert TSwapPool__OutputTooLow(outputAmount,
minOutputAmount);
28 +            revert TSwapPool__OutputTooLow(output, minOutputAmount);
29        }
30
31 -        _swap(inputToken, inputAmount, outputToken, outputAmount);
32 +        _swap(inputToken, inputAmount, outputToken, output);
33    }
```

Info

[I-1] Lack of zero address validation

Check for `address(0)` when assigning values to address state variables.

Findings

- Found in `src/PoolFactory.sol` Line: 43

```
1 +   if (wethToken == address(0)) {
2 +       revert();
3 +   }
4     i_wethToken = wethToken;
```

- Found in `src/TSwapPool.sol` Line: 99

```
1 +   if (wethToken == address(0)) {
```

```
2 +     revert();
3 + }
4     i_wethToken = IERC20(wethToken);
```

- Found in src/TSwapPool.sol Line: 100

```
1 +     if (poolToken == address(0)) {
2 +         revert();
3 +     }
4     i_poolToken = IERC20(poolToken);
```

[I-2] PoolFactory::PoolFactory__PoolDoesNotExist is not used and should be removed

```
1 -     error PoolFactory__PoolDoesNotExist(address tokenAddress);
```

[I-3] PoolFactory::createPool should use .symbol() instead of .name()

```
1
2 -     string memory liquidityTokenSymbol = string.concat(
3 -         "ts",
4 -         IERC20(tokenAddress).name()
5 -     );
6 +     string memory liquidityTokenSymbol = string.concat(
7 +         "ts",
8 +         IERC20(tokenAddress).symbol()
9 +     );
```

[I-4] Event TSwapPool::Swap is missing indexed fields

Each event should use three indexed fields if there are three or more fields, and gas usage is not particularly of concern for the events in question. If there are fewer than three fields, all of the fields should be indexed.

Foundings

- Found in src/TSwapPool.sol Line: 52
- Found in src/TSwapPool.sol Line: 57
- Found in src/TSwapPool.sol Line: 63

[I-5] Function TSwapPool::swapExactInput should be external instead of public because it is not used internally

```

1      function swapExactInput(
2          IERC20 inputToken,
3          uint256 inputAmount,
4          IERC20 outputToken,
5          uint256 minOutputAmount,
6          uint64 deadline
7      )
8  -      public
9  +      external

```

```
3 + uint256 private constant INPUT_AMOUNT = 1e18;
4 + uint256 private constant EXTRA_INCENTIVE = 1
  _000_000_000_000_000_000;
```

Findings

- Found in src/TSwapPool.sol Line: 287

```
1 - uint256 inputAmountMinusFee = inputAmount * 997;
2 + uint256 inputAmountMinusFee = inputAmount * WITHOUT_FEE_MULTIPLIER;
```

- Found in src/TSwapPool.sol Line: 289

```
1 - uint256 denominator = (inputReserves * 1000) + inputAmountMinusFee;
2 + uint256 denominator = (inputReserves * WITH_FEE_MULTIPLIER) +
  inputAmountMinusFee;
```

- Found in src/TSwapPool.sol Line: 307

```
1 return
2 - ((inputReserves * outputAmount) * 10000) /
3 + ((inputReserves * outputAmount) * WITH_FEE_MULTIPLIER) /
4 - ((outputReserves - outputAmount) * 997);
5 + ((outputReserves - outputAmount) * WITHOUT_FEE_MULTIPLIER);
```

- Found in src/TSwapPool.sol Line: 428

```
1 - outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000);
2 + outputToken.safeTransfer(msg.sender, EXTRA_INCENTIVE);
```

- Found in src/TSwapPool.sol Line: 482

```
1 function getPriceOfOneWethInPoolTokens() external view returns (
  uint256) {
2 return
3   getOutputAmountBasedOnInput(
4 -     1e18,
5 +     INPUT_AMOUNT,
6     i_wethToken.balanceOf(address(this)),
7     i_poolToken.balanceOf(address(this))
8   );
9 }
```

- Found in src/TSwapPool.sol Line: 492

```
1 function getPriceOfOnePoolTokenInWeth() external view returns (
  uint256) {
2 return
```

```
3         getOutputAmountBasedOnInput(  
4 -         1e18,  
5 +         INPUT_AMOUNT,  
6         i_poolToken.balanceOf(address(this)),  
7         i_wethToken.balanceOf(address(this))  
8     );  
9 }
```

[I-9] It's better to follow the CEI (Checks, Effects, and Interactions) principle in the else branch of the TSwap::deposit function

```
1     } else {  
2         // This will be the "initial" funding of the protocol. We are  
3         // starting from blank here!  
4         // We just have them send the tokens in, and we mint liquidity  
5         // tokens based on the weth  
6 +         liquidityTokensToMint = wethToDeposit;  
7         _addLiquidityMintAndTransfer(  
8             wethToDeposit,  
9             maximumPoolTokensToDeposit,  
10            wethToDeposit  
11        );  
12 -         liquidityTokensToMint = wethToDeposit;  
13     }
```

Gas

[G-1] Function TSwapPool::_isUnknown can be simplified

```
1     function _isUnknown(IERC20 token) private view returns (bool) {  
2 -         if (token != i_wethToken && token != i_poolToken) {  
3 -             return true;  
4 -         }  
5 -         return false;  
6 +         return token != i_wethToken && token != i_poolToken;  
7     }
```

[G-2] Local variable poolTokenReserves in the function TSwapPool::deposit is not used

```
1     if (totalLiquidityTokenSupply() > 0) {  
2         uint256 wethReserves = i_wethToken.balanceOf(address(this));  
3 -         uint256 poolTokenReserves = i_poolToken.balanceOf(address(this))  
4     };
```